

M. SZYSZKOWICZ (Wrocław)

**A PROCEDURE REALIZING A FOURTH ORDER ONE-STEP METHOD
FOR SOLVING A SYSTEM OF ORDINARY DIFFERENTIAL EQUATIONS
OF THE FORM $y'' = f(x, y, y')$**

1. Procedure declaration. The procedure *sodebis4* solves the initial value problem of the form

$$(1) \quad y_k'' = f_k(x, y_1(x), y_2(x), \dots, y_n(x), y_1'(x), y_2'(x), \dots, y_n'(x)),$$

$$(2) \quad y_k(x_0) = y_{0k},$$

$$(3) \quad y_k'(x_0) = y_{0k} \quad (k = 1, 2, \dots, n)$$

at the points x_1, x_2, \dots

Data:

x — the value of x_0 in (2) and (3);

$x1$ — the value of the argument for which we solve the problem;

eps — the relative error, the given tolerance;

eta — the number which is used instead of zero if the obtained solution is zero or near to zero; this number is used for evaluation of the relative error;

$hmin$ — the least admissible absolute value of the step length;

n — the number of differential equations in (1)-(3);

$y[1:n]$ — the values of the right-hand sides of (2);

$yp[1:n]$ — the values of the right-hand sides of (3).

Results:

x — the value of $x1$;

$y[1:n]$ — the values of the approximate solution $y_k(x1)$
($k = 1, 2, \dots, n$);

$yp[1:n]$ — the values of the approximate solution $y_k'(x1)$
($k = 1, 2, \dots, n$).

Additional parameters:

steph — the label outside of the body of the procedure *sodebis4* to which a jump is made if the absolute value of the step length is smaller than *hmin*; after the jump, x is equal to the value of \tilde{x} ($\tilde{x} < x1$).

for which the approximate solution has a relative error equal to the given one, and $y[1:n]$, $yp[1:n]$ contain the values of this approximate solution;

f — the identifier of the procedure which computes the values of the right-hand sides of (1) and puts them in $d[1:n]$, and which has the following heading:

procedure $f(x, n, y, yp, d)$; **value** x, n ; **real** x ; **integer** n ;
array y, yp, d ;

2. Method used. We use a method of fourth order of the following form:

$$\eta p_{n+1/m}^2 = \eta p_n^5 + \frac{h}{m} f_n^{5,i}, \quad \eta_{n+1/m}^3 = \eta_n^5 + \frac{h}{m} \eta p_n^5 + \frac{h^2}{2m^2} f_n^{5,i},$$

$$\eta p_{n+1/3}^3 = \eta p_n^5 + \frac{h}{3} \left(\frac{6-m}{6} f_n^{5,i} + \frac{m}{6} f_{n+1/m}^{3,2} \right),$$

$$\eta_{n+1/3}^4 = \eta_n^5 + \frac{h}{3} \eta p_n^5 + \frac{h^2}{9} \left(\frac{9-m}{18} f_n^{5,i} + \frac{m}{18} f_{n+1/m}^{3,2} \right),$$

$$\eta p_{n+1/2}^4 = \eta p_n^5 + \frac{h}{8} (f_n^{5,i} + 3f_{n+1/3}^{4,3}),$$

$$\eta_{n+1/2}^4 = \eta_n^5 + \frac{h}{2} \eta p_n^5 + \frac{h^2}{16} (f_n^{5,i} + f_{n+1/3}^{4,3}),$$

$$\eta p_{n+1}^4 = \eta p_n^5 + \frac{h}{2} (f_n^{5,i} - 3f_{n+1/3}^{4,3} + 4f_{n+1/2}^{4,4}), \quad \eta_{n+1}^4 = \eta_n^5 + h \eta p_n^5 + \frac{h^2}{2} f_{n+1/3}^{4,3},$$

$$\eta_{n+1}^5 = \eta_n^5 + h \eta p_n^5 + \frac{h^2}{6} (f_n^{5,i} + 2f_{n+1/2}^{4,4}),$$

$$\eta p_{n+1}^5 = \eta p_n^5 + \frac{h}{6} (f_n^{5,i} + 4f_{n+1/2}^{4,4} + f_{n+1}^{5,4}),$$

where η_{n+a}^k is the approximate value of the solution of (1)-(3) at the point $x_n + ah$ with local error $O(h^k)$, ηp_{n+a}^k is the approximate value of the first derivative of the solution of (1)-(3) at the point $x_n + ah$ with local error $O(h^k)$ and $f_{n+a}^{i,j} = f(x_n + ah, \eta_{n+a}^i, \eta p_{n+a}^j)$. The parameter m ($m > 0$) was assumed to be equal to 4 and $i = 5$.

This method was given by Bobkov [2], p. 38-42. Twofold application of the method with step $h/2$ allows us to obtain the values with step h without evaluation of the function f .

In the algorithm, 9 evaluations of f in one step are made. The method of control of the step of integration described in [1] is used.

```

procedure sodebis4(x,x1,eps,eta,hmin,n,y,yp,steph,f);
  value x1,eps,eta,hmin,n;
  real x,x1,eps,eta,hmin;
  integer n;
  array y,yp;
  label steph;
  procedure f;
  begin
    real a1,b,b1,h,hh,hk,h1,h2,w,w1,w2,w3,w4;
    integer i;
    Boolean last;
    array d,d1,d2,d3,yp2,yp3,y2,y3[1:n];
    procedure stepbis4(x,d,y,ya,yp,yb);
      value x;
      real x;
      array d,y,ya,yp,yb;
      begin
        w:=hh*.25;
        w1:=hk*.03125;
        for i:=1 step 1 until n do
          begin
            w2:=d[i];
            w3:=yp[i];
            yb[i]:=w3+w*w2;
            ya[i]:=y[i]+w*w3+w1*w2
          end i;
        h1:=hh*.333333333333;
        h2:=hk*.006172777777;
        b:=2.0;
        a1:=5.0;

```

M. Szyszkowicz

```
b1:=4.0;
for w1:=hh*.111111111111, hh*.125 do
  begin
    f(x+w,n,ya,yb,d2);
    for i:=1 step 1 until n do
      begin
        w2:=d2[i];
        w3:=yp[i];
        w4:=d[i];
        yb[i]:=w3+w1*(w4+w2*b);
        ya[i]:=y[i]+h1*w3+h2*(a1*w4+b1*w2)
      end i;
    w:=h1;
    h1:=hh*.5;
    h2:=hk*.0625;
    b:=3.0;
    a1:=b1:=1.0
  end w1;
f(x+h1,n,ya,yb,d3);
h2:=hk*.166666666666;
for i:=1 step 1 until n do
  begin
    w2:=d[i];
    w3:=d3[i];
    w4:=yp[i];
    yb[i]:=w4+h1*(w2-3.0*d2[i]+4.0*w3);
    ya[i]:=y[i]+hh*w4+h2*(w2+w3+w3)
  end i;
f(x+hh,n,ya,yb,d2);
h1:=hh*.166666666666;
```

```

    for i:=1 step 1 until n do
        yb[i]:=yp[i]+h1*(d[i]+4.0*d3[i]+d2[i])
    end stepbis4;
eps:=.0333333333333/eps;
h:=x1-x;
last:=true;
f(x,n,y,yp,d);
conth:
    hh:=h*.5;
    hk:=hh*hh;
    stepbis4(x,d,y,y2,yp,yp2);
    for i:=1 step 1 until n do
        d1[i]:=d2[i];
    stepbis4(x+hh,d1,y2,y3,yp2,yp3);
    w:=.0;
    h1:=h*.166666666666;
    h2:=h*h*.166666666666;
    for i:=1 step 1 until n do
        begin
            w2:=d[i];
            w1:=d1[i];
            w3:=yp[i];
            a1:=y3[i];
            w4:=yp3[i];
            b:=a1-y[i]-h*w3-h2*(w2+w1+w1);
            b1:=w4-w3-h1*(w2+4.0*w1+d2[i]);
            w1:=y3[i]:=a1+.066666666666*b;
            w2:=yp3[i]:=w4+.066666666666*b1;
            w1:=abs(w1);
            w2:=abs(w2);

```

```

b1:=abs(b1);
b:=abs(b);
if w1<eta
  then w1:=eta;
if w2<eta
  then w2:=eta;
w1:=b/w1;
w2:=b1/w2;
if w1>w
  then w:=w1;
if w2>w
  then w:=w2
end i;
w:=if w=.0 then eta else 1.25*(w*eps)↑.2;
hh:=h/w;
if w>1.25
  then
  begin
    if abs(hh)<hmin
      then go to steph;
    last:=false
  end w>1.25
  else
  begin
    x:=x+h;
    for i:=1 step 1 until n do
      begin
        y[i]:=y3[i];
        yp[i]:=yp3[i]
      end i;
  end

```

```

if last
  then go to endp;
  f(x,n,y,yp,d);
  w:=x1-x;
  if (w-hh)*h<0
    then
      begin
        hh:=w;
        last:=true
      end(w-hh)*h<0
    end w<1.25;
  h:=hh;
  go to conth;
endp:
end sodebis4

```

3. Certification. The procedure *sodebis4* has been verified on the Odra 1204 computer for many examples of the initial value problem. Some of them are presented here.

Examples.

$$(A) \quad \begin{cases} y_1'' = -y_2'/y_2^2, & y_1(0) = 1, & y_1'(0) = 1, \\ y_2'' = y_1'/y_1^2, & y_2(0) = 1, & y_2'(0) = -1 \end{cases}$$

with the exact solution $y_1 = e^x$, $y_2 = e^{-x}$.

$$(B) \quad \begin{cases} y_1'' = y_1, & y_1(0) = 1, & y_1'(0) = 1, \\ y_2'' = -y_2, & y_2(0) = 0, & y_2'(0) = -1 \end{cases}$$

with the exact solution $y_1 = e^x$, $y_2 = \sin x$.

$$(C) \quad \begin{cases} y_1'' = y_1/4, & y_1(0) = 1, & y_1'(0) = -1/2, \\ y_2'' = (1+x^2)y_2, & y_2(0) = 1, & y_2'(0) = 0 \end{cases}$$

with the exact solution $y_1 = e^{-x/2}$, $y_2 = e^{x^2/2}$.

The results, obtained for $eps = eta$ and $hmin = 10^{-15}$, are given below. As initial values, the exact solutions at the points .5, 1.0, 1.5 were used. The relative errors $(\eta - y)/y$ and the numbers of evaluations of the function f (denoted by $[f]$) are also given.

Results obtained for problem (A)

x	$eps = 10^{-3}$	$[f]$	$eps = 10^{-6}$	$[f]$	$eps = 10^{-9}$	$[f]$
.5	-1.0_{10}^{-6} -1.2_{10}^{-6}	9	-2.7_{10}^{-7} -4.3_{10}^{-7}	26	-1.1_{10}^{-9} -2.8_{10}^{-9}	62
1.0	-1.0_{10}^{-6} -1.2_{10}^{-6}	9	-2.7_{10}^{-7} -4.2_{10}^{-7}	26	-1.1_{10}^{-9} -2.9_{10}^{-9}	62
1.5	-1.0_{10}^{-6} -1.2_{10}^{-6}	9	-2.7_{10}^{-7} -4.2_{10}^{-7}	26	-1.1_{10}^{-9} -2.7_{10}^{-9}	62
10.0	(¹)		1.6_{10}^{-2} -3.2_{10}^{-2}	224	8.2_{10}^{-5} -1.6_{10}^{-4}	827

Results obtained for problem (B)

x	$eps = 10^{-3}$	$[f]$	$eps = 10^{-6}$	$[f]$	$eps = 10^{-9}$	$[f]$
.5	1.5_{10}^{-7} 9.7_{10}^{-8}	9	2.6_{10}^{-8} 1.1_{10}^{-8}	26	1.4_{10}^{-10} -4.5_{10}^{-11}	62
1.0	1.5_{10}^{-7} -1.2_{10}^{-7}	9	5.2_{10}^{-8} -4.2_{10}^{-8}	26	2.5_{10}^{-10} -2.2_{10}^{-10}	62
1.5	1.5_{10}^{-7} -2.2_{10}^{-7}	9	1.9_{10}^{-8} -2.7_{10}^{-8}	26	1.8_{10}^{-10} -2.9_{10}^{-10}	62
10.0	1.4_{10}^{-4} 1.2_{10}^{-3}	70	1.8_{10}^{-6} 6.7_{10}^{-6}	291	7.5_{10}^{-9} 1.7_{10}^{-8}	1095

Results obtained for problem (C)

x	$eps = 10^{-3}$	$[f]$	$eps = 10^{-6}$	$[f]$	$eps = 10^{-9}$	$[f]$
.5	6.5_{10}^{-9} 3.3_{10}^{-7}	9	3.1_{10}^{-10} 4.2_{10}^{-8}	35	-3.7_{10}^{-11} 1.5_{10}^{-10}	115
1.0	6.5_{10}^{-9} -1.0_{10}^{-6}	9	2.8_{10}^{-10} 4.6_{10}^{-8}	35	-2.3_{10}^{-11} 2.2_{10}^{-10}	98
1.5	6.5_{10}^{-9} -3.0_{10}^{-6}	9	1.3_{10}^{-10} 7.8_{10}^{-8}	35	6.1_{10}^{-11} 4.3_{10}^{-10}	1107
10.0	1.2_{10}^{-3} -1.9_{10}^{-4}	267	1.5_{10}^{-6} 1.5_{10}^{-5}	1195	2.0_{10}^{-7} 6.8_{10}^{-8}	4912

(¹) The determined step length was smaller than $hmin$.

References

- [1] J. S. Chomicz, A. Olejniczak and M. Szyszkowicz, *A method for finding the step size of integration of a system of ordinary differential equations*, *Zastos. Mat.* 17 (1983), p. 645-654.
- [2] V. I. Krylov, V. V. Bobkov and P. I. Monastyrnyĭ (В. И. Крылов, В. В. Бобков и П. И. Монастырныĭ), *Вычислительные методы*, т. 2, Москва 1977.

INSTITUTE OF COMPUTER SCIENCE
UNIVERSITY OF WROCLAW
50-384 WROCLAW

Received on 2. 6. 1980

